

□ Prove di calcolo tensoriale

Provo a realizzare un documento wxm per il calcolo di tutte le grandezze derivate da un assegnato tensore metrico. In questo caso uso una dipendenza funzionale del tutto generale ossia uso una generica funzione $\psi(r)$ dipendente dalla sola coordinata r . In questo modo posso trattare sia il buco nero neutro (Schwarzschild) che quello carico (Reissner e Nordstrom) ma anche un qualsiasi altro tipo di buco nero che produca il tensore metrico che ho imposto.

```
(%i1) (if atom(lg) then load(ctensor));
(%o1)
C:/Programmi/Maxima-5.20.1/share/maxima/5.20.1/share/tensor/ctensor.mac
```

Pulisco le variabili se per caso fossero state inizializzate in modo non corretto.

```
(%i2) init_ctensor();
(%o2) done
```

Specifico quali variabili debbono essere considerate variabili indipendenti. Ovviamente uso i classici nomi in geometria sferica ovvero t, r, θ e ϕ .

```
(%i3) ct_coords: [t,r,theta,phi];
(%o3) [t, r, theta, phi]
```

Dichiaro che il simbolo ψ deve essere considerato una funzione della variabile r .

```
(%i4) depends(ψ, r);
(%o4) [ψ(r)]
```

La libreria ctensor si aspetta di trovare lg definita come una matrice quadrata di ordine quattro interpretata come il tensore metrico covariante.

```
(%i5) lg:matrix(
  [1+ψ,0,0,0],
  [0,-1/(1+ψ),0,0],
  [0,0,-r^2,0],
  [0,0,0,-(r*sin(theta))^2]);
(%o5) [
  ψ+1  0  0  0
  0  -1/(ψ+1)  0  0
  0  0  -r^2  0
  0  0  0  -r^2 sin(theta)^2]
```

```
(%i6) lg;
```

$$\begin{matrix}
 & \left[\begin{array}{cccc}
 \text{psi} + 1 & 0 & 0 & 0 \\
 0 & \frac{1}{\text{psi} + 1} & 0 & 0 \\
 0 & 0 & -r^2 & 0 \\
 0 & 0 & 0 & -r^2 \sin(\text{theta})^2
 \end{array} \right] \\
 (%o6) & & &
 \end{matrix}$$

Con la funzione cmetric() creo il tensore metrico controvariante indicato come ug e definito come la matrice inversa di lg.

```
(%i7) cmetric();
(%o7) done
```

```
(%i8) ug;
```

$$\begin{matrix}
 & \left[\begin{array}{cccc}
 \frac{1}{\text{psi} + 1} & 0 & 0 & 0 \\
 0 & -\text{psi} - 1 & 0 & 0 \\
 0 & 0 & -\frac{1}{r^2} & 0 \\
 0 & 0 & 0 & -\frac{1}{r^2 \sin(\text{theta})^2}
 \end{array} \right] \\
 (%o8) & & &
 \end{matrix}$$

A questo punto calcolo i simboli di Christoffel. Specificando all li visualizzo tutti ossia quelli di prima specie contenuti nell'array lcs, sia quelli di seconda specie contenuti nell'array ucs. Notare che la numerazione delle array in massima parte da 0 ma qui gli indici 0 non sono utilizzati. Da notare anche che l'ordine e' invertito rispetto a quello tradizionale per cui sono simmetrici tra loro il primo ed is secondo indice e non, come si trova su molti libri, il secondo con il terzo.

```

(%i9) christof(all);
(%t9)  $lcs_{1,1,2} = -\frac{\psi_r}{2}$ 
(%t10)  $lcs_{1,2,1} = \frac{\psi_r}{2}$ 
(%t11)  $lcs_{2,2,2} = \frac{\psi_r}{2(\psi+1)^2}$ 
(%t12)  $lcs_{2,3,3} = -r$ 
(%t13)  $lcs_{2,4,4} = -r \sin(\theta)^2$ 
(%t14)  $lcs_{3,3,2} = r$ 
(%t15)  $lcs_{3,4,4} = -r^2 \cos(\theta) \sin(\theta)$ 
(%t16)  $lcs_{4,4,2} = r \sin(\theta)^2$ 
(%t17)  $lcs_{4,4,3} = r^2 \cos(\theta) \sin(\theta)$ 
(%t18)  $mcs_{1,1,2} = \frac{(\psi+1)(\psi_r)}{2}$ 
(%t19)  $mcs_{1,2,1} = \frac{\psi_r}{2\psi+2}$ 
(%t20)  $mcs_{2,2,2} = -\frac{\psi_r}{2\psi+2}$ 
(%t21)  $mcs_{2,3,3} = \frac{1}{r}$ 
(%t22)  $mcs_{2,4,4} = \frac{1}{r}$ 
(%t23)  $mcs_{3,3,2} = (-\psi-1)r$ 
(%t24)  $mcs_{3,4,4} = \frac{\cos(\theta)}{\sin(\theta)}$ 
(%t25)  $mcs_{4,4,2} = (-\psi-1)r \sin(\theta)^2$ 
(%t26)  $mcs_{4,4,3} = -\cos(\theta) \sin(\theta)$ 
(%o26) done

```

Calcolo il tensore di Riemann che opportunamente contratto mi fornisce il tensore di Ricci.

```
(%i27) riemann(true);
(%t27)  $riem_{1,2,1,2} = -\frac{(psi+1)(psi_{rr})}{2} + \frac{(psi+1)(psi_r)^2}{2 psi+2} - \frac{(psi_r)^2}{2}$ 
(%t28)  $riem_{1,3,1,3} = -\frac{(psi+1)(psi_r)}{2 r}$ 
(%t29)  $riem_{1,4,1,4} = -\frac{(psi+1)(psi_r)}{2 r}$ 
(%t30)  $riem_{2,2,1,1} = -\frac{psi_{rr}}{2 psi+2}$ 
(%t31)  $riem_{2,3,2,3} = \frac{psi_r}{(2 psi+2) r}$ 
(%t32)  $riem_{2,4,2,4} = \frac{psi_r}{(2 psi+2) r}$ 
(%t33)  $riem_{3,3,1,1} = -\frac{(psi_r) r}{2}$ 
(%t34)  $riem_{3,3,2,2} = -\frac{(psi_r) r}{2}$ 
(%t35)  $riem_{3,4,3,4} = psi$ 
(%t36)  $riem_{4,4,1,1} = -\frac{(psi_r) r \sin(theta)^2}{2}$ 
(%t37)  $riem_{4,4,2,2} = -\frac{(psi_r) r \sin(theta)^2}{2}$ 
(%t38)  $riem_{4,4,3,3} = -psi \sin(theta)^2$ 
(%o38) done
```

Calcolo il tensore di Ricci.

```
(%i39) ricci(true);
(%t39)  $ric_{1,1} = \frac{(psi+1)(psi_r)}{r} + \frac{(psi+1)(psi_{rr})}{2} - \frac{(psi+1)(psi_r)^2}{2 psi+2} + \frac{(psi_r)^2}{2}$ 
(%t40)  $ric_{2,2} = -\frac{2(psi_r)}{(2 psi+2) r} - \frac{psi_{rr}}{2 psi+2}$ 
(%t41)  $ric_{3,3} = -(psi_r) r - psi$ 
(%t42)  $ric_{4,4} = -(psi_r) r \sin(theta)^2 + (-psi - 1) \sin(theta)^2 + \sin(theta)^2$ 
(%o42) done
```

Lo visualizzo sotto forma di lista semplificandolo.

```
(%i43) ratsimp(listarray(ric));
(%o43) [ $\frac{(psi+1)(psi_{rr}) r + (2 psi+2)(psi_r)}{2 r}$ , 0, 0, 0, 0,  $-\frac{(psi_{rr}) r + 2(psi_r)}{(2 psi+2) r}$ , 0, 0, 0, 0,  $-(psi_r) r - psi$ , 0, 0, 0, 0,  $(-psi - 1) \sin(theta)^2 + \sin(theta)^2$ ]
```

⌈ Ora carico l'array ric in una matrice che chiamo mat_ricci

```
(%i44) mat_ricci: ratsimp( matrix(
    [ ric[1,1],ric[1,2],ric[1,3],ric[1,4]],
    [ ric[2,1],ric[2,2],ric[2,3],ric[2,4]],
    [ ric[3,1],ric[3,2],ric[3,3],ric[3,4]],
    [ ric[4,1],ric[4,2],ric[4,3],ric[4,4]]));
```

(%o44)

$$\begin{bmatrix} \frac{(\psi+1)(\psi_{r,r})r+(2\psi+2)(\psi_r)}{2r} & 0 & 0 & 0 \\ 0 & -\frac{(\psi_{r,r})r+2(\psi_r)}{(2\psi+2)r} & 0 & 0 \\ 0 & 0 & -(\psi_r)r-\psi & 0 \\ 0 & 0 & 0 & (-(\psi_r)r-\psi)\sin(\theta)^2 \end{bmatrix}$$

⌈ Invece dei simboli di derivata uso u0 per la funzione non derivata, u1 per la sua derivata prima rispetto ad r e u2 per la sua derivata seconda. Faccio questo per usare piu' facilmente la funzione subst. Chiamo il risultato mu_ricci.

```
(%i45) (u0:"u0",u1:"u1",u2:"u2");
(%o45) u2
```

(%i46) mu_ricci: subst(u0,psi,subst(u2,diff(psi,r,2),subst(u1,diff(psi,r),ma

(%o46)

$$\begin{bmatrix} \frac{(u0+1)u2r+(2u0+2)u1}{2r} & 0 & 0 & 0 \\ 0 & -\frac{u2r+2u1}{(2u0+2)r} & 0 & 0 \\ 0 & 0 & -u1r-u0 & 0 \\ 0 & 0 & 0 & (-u1r-u0)\sin(\theta)^2 \end{bmatrix}$$

⌈ Ora mostro che se assumo che u0 valga -2*m/r ed u1 ed u2 di conseguenza, il tensore di Ricci si annulla ossia ho la soluzione del buco nero neutro ovvero la soluzione di Schwarzschild. Se invece q non è nullo e pongo u0 = -2*m/r+q^2/r^2 ottengo la soluzione di Reissner e Nordstrom.

```
(%i47) v0:-2*m/r+q^2/r^2;
(%o47) \frac{q^2}{r^2} - \frac{2m}{r}
```

```
(%i48) v1:diff(v0,r);
(%o48) \frac{2m}{r^2} - \frac{2q^2}{r^3}
```

(%i49) v2:diff(v0,r,2);

(%o49) $\frac{6 q^2}{r^4} - \frac{4 m}{r^3}$

Sostituisco v0, v1 e v2 ad u0, u1 ed u2.

(%i50) ratsimp(subst(v0,u0,subst(v1,u1,subst(v2,u2,mu_ricci))));

(%o50)
$$\begin{bmatrix} \frac{q^2 r^2 - 2 m q^2 r + q^4}{r^6} & 0 & 0 & 0 \\ 0 & -\frac{q^2}{r^4 - 2 m r^3 + q^2 r^2} & 0 & 0 \\ 0 & 0 & \frac{q^2}{r^2} & 0 \\ 0 & 0 & 0 & \frac{q^2 \sin(\theta)^2}{r^2} \end{bmatrix}$$

...etc...